Добрый день всем!

Есть отчёт на скд.

Вывожу суммуКТ по кварталам за год из регистра бухгалтерии (скрин1) по основным средствам.
Необходимо вывести суммуКНалогу основногосредства так же по кварталам которая хранится в непереодическом регистре бухгалтерии.

Встрял на моменте связи "Периода" у РС с периодом-кварталам регистра бух. (скрин2).

Собственно как правильно связать периода РС с каждым квартлам РБ ?
---
```1c
    ВЫБРАТЬ
    НАЧАЛОПЕРИОДА(РасчетНалогаНаИмущество.ПериодРасчета, КВАРТАЛ) КАК Квартал,
    РасчетНалогаНаИмущество.ОсновноеСредство КАК ОсновноеСредство,
    СУММА(РасчетНалогаНаИмущество.СуммаНалогаКУплате) КАК СуммаНалогаКУплате
    ИЗ
    РегистрСведений.РасчетНалогаНаИмущество КАК РасчетНалогаНаИмущество

    СГРУППИРОВАТЬ ПО
    НАЧАЛОПЕРИОДА(РасчетНалогаНаИмущество.ПериодРасчета, КВАРТАЛ),
    РасчетНалогаНаИмущество.ОсновноеСредство
```

Получилось что-то такое, а далее соединяйте по кварталу и основному средству.
Разные суммы в консоли и СКД могут быть из-за настройки измерений в макете компоновки и установленных отборов, т.к. СКД может оптимизировать отборы из-за чего бывает путаница. Да и не забудьте на РС РасчетНалогаНаИмущество наложить отбор по основному средству, чтобы не соединяться со всем регистром.
---
Мб стоит так поступить. Связать свои параметры со стандартным периодом
---
Выйдет так, что мне придётся добавить на форму ещё возможность выбирать дату начала и дата конца.
Но у меня сейчас выбирают только год. Мне надо ограничиться этим полем
---
Периодичность "Квартал" - это первое число месяца начала квартала. Для упрощения понимания в каждой временной таблице создайте отдельную колонку в которой будете хранить преобразованную дату, что-то типа:

```1c
    ДАТАВРЕМЯ(ГОД(Период), МесяцКвартала, 1)
```

МесяцКвартала можно вычислить так:

```1c
    ВЫБОР КВАРТАЛ(Период)
    КОГДА 1
    ТОГДА 1
    КОГДА 2
    ТОГДА 4
    КОГДА 3
    ТОГДА 7
    ИНАЧЕ 10
    КОНЕЦ
```
---
```1c
    ВЫБРАТЬ
    ХозрасчетныйОбороты.Период КАК Квартал,
    ХозрасчетныйОбороты.Субконто1 КАК ОсновноеСредство,
    ХозрасчетныйОбороты.СуммаОборотКт КАК Сумма
    ПОМЕСТИТЬ Хозрасчетный
    ИЗ
    РегистрБухгалтерии.Хозрасчетный.Обороты(&НачПериода, &КонПериода, Квартал, , ЗНАЧЕНИЕ(ПланВидовХарактеристик.ВидыСубконтоХозрасчетные.ОсновныеСредства), , , ) КАК ХозрасчетныйОбороты

    ИНДЕКСИРОВАТЬ ПО
    ОсновноеСредство
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ РАЗРЕШЕННЫЕ
    ПервоначальныеСведенияОСБухгалтерскийУчетСрезПоследних.ОсновноеСредство КАК ОсновноеСредство,
    ПервоначальныеСведенияОСБухгалтерскийУчетСрезПоследних.ИнвентарныйНомер КАК ИнвентарныйНомер,
    ПервоначальныеСведенияОСБухгалтерскийУчетСрезПоследних.ПервоначальнаяСтоимость КАК ПервоначальнаяСтоимость,
    ПервоначальныеСведенияОСБухгалтерскийУчетСрезПоследних.ПорядокПогашенияСтоимости КАК ПорядокПогашенияСтоимости,
    ВЫБОР
    КОГДА ПервоначальныеСведенияОСБухгалтерскийУчетСрезПоследних.ПорядокПогашенияСтоимости = ЗНАЧЕНИЕ(Перечисление.ПорядокПогашенияСтоимостиОС.НачислениеАмортизации)
    ТОГДА ПервоначальныеСведенияОСБухгалтерскийУчетСрезПоследних.СпособНачисленияАмортизации
    ИНАЧЕ ПервоначальныеСведенияОСБухгалтерскийУчетСрезПоследних.ПорядокПогашенияСтоимости
    КОНЕЦ КАК СпособНачисленияАмортизации
    ПОМЕСТИТЬ ПервоначальныеСведенияОСБухгалтерскийУчетКонПериода
    ИЗ
    РегистрСведений.ПервоначальныеСведенияОСБухгалтерскийУчет.СрезПоследних(&КонПериода, ) КАК ПервоначальныеСведенияОСБухгалтерскийУчетСрезПоследних

    ИНДЕКСИРОВАТЬ ПО
    ОсновноеСредство
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    Хозрасчетный.Квартал КАК Квартал,
    Хозрасчетный.ОсновноеСредство КАК ОсновноеСредство,
    ПервоначальныеСведенияОСБухгалтерскийУчетКонПериода.ИнвентарныйНомер КАК ИнвентарныйНомер,
    Хозрасчетный.Сумма КАК Сумма,
    РасчетНалогаНаИмущество.СуммаНалогаКУплате КАК СуммаНалогаКУплате
    ИЗ
    Хозрасчетный КАК Хозрасчетный
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ ПервоначальныеСведенияОСБухгалтерскийУчетКонПериода КАК ПервоначальныеСведенияОСБухгалтерскийУчетКонПериода
    ПО Хозрасчетный.ОсновноеСредство = ПервоначальныеСведенияОСБухгалтерскийУчетКонПериода.ОсновноеСредство
    ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.РасчетНалогаНаИмущество КАК РасчетНалогаНаИмущество
    ПО Хозрасчетный.ОсновноеСредство = РасчетНалогаНаИмущество.ОсновноеСредство

    УПОРЯДОЧИТЬ ПО
    Квартал
```

Собственно вот запрос.
Идею Вашу понял, но не пойму как в запросе это добавить корректно ?
---
```1c
    ВЫБРАТЬ
    НАЧАЛОПЕРИОДА(РасчетНалогаНаИмущество.ПериодРасчета, КВАРТАЛ) КАК Квартал,
    РасчетНалогаНаИмущество.ОсновноеСредство КАК ОсновноеСредство,
    СУММА(РасчетНалогаНаИмущество.СуммаНалогаКУплате) КАК СуммаНалогаКУплате
    ИЗ
    РегистрСведений.РасчетНалогаНаИмущество КАК РасчетНалогаНаИмущество

    СГРУППИРОВАТЬ ПО
    НАЧАЛОПЕРИОДА(РасчетНалогаНаИмущество.ПериодРасчета, КВАРТАЛ),
    РасчетНалогаНаИмущество.ОсновноеСредство
```

Получилось что-то такое, а далее соединяйте по кварталу и основному средству.
Разные суммы в консоли и СКД могут быть из-за настройки измерений в макете компоновки и установленных отборов, т.к. СКД может оптимизировать отборы из-за чего бывает путаница. Да и не забудьте на РС РасчетНалогаНаИмущество наложить отбор по основному средству, чтобы не соединяться со всем регистром.
---
Всё получилось! Спасибо,вы крутой!

По поводу отбору по основному средству не совсем понял правда. Имеете ввиду в ГДЕ наложить отбор на регистр ?
Но я подразумеваю что пользователь будет пользоваться типовым отбором от скд.Где уже сможет сам выбирать основные средства.
---
Спасибо)
Я имел ввиду, что вы сначала выбираете обороты по всем ОС за период, а потом соединяете эту таблицу с регистром сведений напрямую. Проблема здесь в том, что в оборотах у вас может быть 10 основных средств, а в РС - 1000 и это будет медленно.

```1c
    ВЫБРАТЬ
    ХозрасчетныйОбороты.Период КАК Квартал,
    ХозрасчетныйОбороты.Субконто1 КАК ОсновноеСредство,
    ХозрасчетныйОбороты.СуммаОборотКт КАК Сумма
    ПОМЕСТИТЬ Хозрасчетный
    ИЗ
    РегистрБухгалтерии.Хозрасчетный.Обороты(&НачПериода, &КонПериода, Квартал, , ЗНАЧЕНИЕ(ПланВидовХарактеристик.ВидыСубконтоХозрасчетные.ОсновныеСредства), , , ) КАК ХозрасчетныйОбороты

    ИНДЕКСИРОВАТЬ ПО
    ОсновноеСредство
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    НАЧАЛОПЕРИОДА(РасчетНалогаНаИмущество.ПериодРасчета, КВАРТАЛ) КАК Квартал,
    РасчетНалогаНаИмущество.ОсновноеСредство КАК ОсновноеСредство,
    СУММА(РасчетНалогаНаИмущество.СуммаНалогаКУплате) КАК СуммаНалогаКУплате
    ИЗ
    РегистрСведений.РасчетНалогаНаИмущество КАК РасчетНалогаНаИмущество
    ГДЕ
    РасчетНалогаНаИмущество.ПериодРасчета МЕЖДУ &НачПериода И &КонПериода
    И РасчетНалогаНаИмущество.ОсновноеСредство В
    (ВЫБРАТЬ
    Хозрасчетный.ОсновноеСредство КАК ОсновноеСредство
    ИЗ
    Хозрасчетный КАК Хозрасчетный)

    СГРУППИРОВАТЬ ПО
    НАЧАЛОПЕРИОДА(РасчетНалогаНаИмущество.ПериодРасчета, КВАРТАЛ),
    РасчетНалогаНаИмущество.ОсновноеСредство
```

Также если у вас сложный запрос с большим количеством временных таблиц, где происходят какие-либо расчеты и в них всех есть поле одинаковым наименованием, то накладывая отбор СКД при преобразовании схемы в запрос, наложит этот отбор не только на результирующую таблицу запроса, но и на все временные, в есть поле отбора.
---
Сейчас обратил внимание, после соединения кварталов пропал 4 квартал при выводе очтета.почему так ?..

```1c
    ВЫБРАТЬ
    ХозрасчетныйОбороты.Период КАК Квартал,
    ХозрасчетныйОбороты.Субконто1 КАК ОсновноеСредство,
    ХозрасчетныйОбороты.СуммаОборотКт КАК Сумма
    ПОМЕСТИТЬ Хозрасчетный
    ИЗ
    РегистрБухгалтерии.Хозрасчетный.Обороты(&НачПериода, &КонПериода, Квартал, , ЗНАЧЕНИЕ(ПланВидовХарактеристик.ВидыСубконтоХозрасчетные.ОсновныеСредства), , , ) КАК ХозрасчетныйОбороты

    ИНДЕКСИРОВАТЬ ПО
    ОсновноеСредство
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ РАЗРЕШЕННЫЕ
    ПервоначальныеСведенияОСБухгалтерскийУчетСрезПоследних.ОсновноеСредство КАК ОсновноеСредство,
    ПервоначальныеСведенияОСБухгалтерскийУчетСрезПоследних.ИнвентарныйНомер КАК ИнвентарныйНомер,
    ПервоначальныеСведенияОСБухгалтерскийУчетСрезПоследних.ПервоначальнаяСтоимость КАК ПервоначальнаяСтоимость,
    ПервоначальныеСведенияОСБухгалтерскийУчетСрезПоследних.ПорядокПогашенияСтоимости КАК ПорядокПогашенияСтоимости,
    ВЫБОР
    КОГДА ПервоначальныеСведенияОСБухгалтерскийУчетСрезПоследних.ПорядокПогашенияСтоимости = ЗНАЧЕНИЕ(Перечисление.ПорядокПогашенияСтоимостиОС.НачислениеАмортизации)
    ТОГДА ПервоначальныеСведенияОСБухгалтерскийУчетСрезПоследних.СпособНачисленияАмортизации
    ИНАЧЕ ПервоначальныеСведенияОСБухгалтерскийУчетСрезПоследних.ПорядокПогашенияСтоимости
    КОНЕЦ КАК СпособНачисленияАмортизации
    ПОМЕСТИТЬ ПервоначальныеСведенияОСБухгалтерскийУчетКонПериода
    ИЗ
    РегистрСведений.ПервоначальныеСведенияОСБухгалтерскийУчет.СрезПоследних(&КонПериода, {(Организация).*, (ОсновноеСредство).*}) КАК ПервоначальныеСведенияОСБухгалтерскийУчетСрезПоследних

    ИНДЕКСИРОВАТЬ ПО
    ОсновноеСредство
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    НАЧАЛОПЕРИОДА(РасчетНалогаНаИмущество.ПериодРасчета, КВАРТАЛ) КАК Квартал,
    РасчетНалогаНаИмущество.ОсновноеСредство КАК ОсновноеСредство,
    СУММА(РасчетНалогаНаИмущество.СуммаНалогаКУплате) КАК СуммаНалогаКУплате
    ПОМЕСТИТЬ Налог
    ИЗ
    РегистрСведений.РасчетНалогаНаИмущество КАК РасчетНалогаНаИмущество
    ГДЕ
    РасчетНалогаНаИмущество.ПериодРасчета МЕЖДУ &НачПериода И &КонПериода
    И РасчетНалогаНаИмущество.ОсновноеСредство В
    (ВЫБРАТЬ
    Хозрасчетный.ОсновноеСредство КАК ОсновноеСредство
    ИЗ
    Хозрасчетный КАК Хозрасчетный)

    СГРУППИРОВАТЬ ПО
    РасчетНалогаНаИмущество.ОсновноеСредство,
    НАЧАЛОПЕРИОДА(РасчетНалогаНаИмущество.ПериодРасчета, КВАРТАЛ)
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    Хозрасчетный.Квартал КАК Квартал,
    Хозрасчетный.ОсновноеСредство КАК ОсновноеСредство,
    ПервоначальныеСведенияОСБухгалтерскийУчетКонПериода.ИнвентарныйНомер КАК ИнвентарныйНомер,
    Хозрасчетный.Сумма КАК Сумма,
    Налог.СуммаНалогаКУплате КАК СуммаНалогаКУплате
    ИЗ
    Хозрасчетный КАК Хозрасчетный
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ ПервоначальныеСведенияОСБухгалтерскийУчетКонПериода КАК ПервоначальныеСведенияОСБухгалтерскийУчетКонПериода
    ПО Хозрасчетный.ОсновноеСредство = ПервоначальныеСведенияОСБухгалтерскийУчетКонПериода.ОсновноеСредство
    ЛЕВОЕ СОЕДИНЕНИЕ Налог КАК Налог
    ПО Хозрасчетный.Квартал = Налог.Квартал
    И Хозрасчетный.ОсновноеСредство = Налог.ОсновноеСредство

    УПОРЯДОЧИТЬ ПО
    Квартал
```
---
У вас во временной таблице "Хозрасчетный" есть данные по 4 кварталу?
Если них нет, то из временной таблицы "Налог" они тоже не попадут, т.к. соединение ЛЕВОЕ.
---
В том то и дело есть. Пока не соединил они отображались.. странно

Upd: стоп, они были, но вдруг пропали из регистра.хм пока вопрос отпал тогда.

То есть ли данные либо в хозрасчетном за 4 квартал появятся, либо основное средство попадет в периодрасчета в РС "РасчетНалогаНаИмущество" то 4 квартал отобразится ?
---
Тут нужно смотреть данные.